
Public PROBABILITY(0,0) As Number ; [number of trials][number of ups] Pr(success) = PROB_UP.
Public MAX_TRIALS As Number
 
Private p As Number
Private q As Number
Private NumTrials As Number
Private NumTrialsPlusOne As Number
Private NumSuccesses As Number
Private Mode As Number
Private LeftIdx As Number
Private RightIdx As Number
Private Total As Number
 
Private Initialized As Number
 
If Initialized = 0 Then
  If MOD_POINT_NO > 1 Then
    ERROR("!!! *** Error! It appears that Parallel Model Point Processing is enabled.  Please disable. *** !!!")
  EndIf
 
  MAX_TRIALS := ROUND_UP((RETIREMENT_AGE_YEARS + 1) * PERIODS_PER_YEAR, 0)
  ARRAY_INITIALISE(PROBABILITY, MAX_TRIALS + 1, MAX_TRIALS + 1)
  p := PROB_DOWN
  q := PROB_UP
 
  ; Save the heavy machinery for harder problems.
  PROBABILITY(0, 0) := 1
  PROBABILITY(1, 0) := q
  PROBABILITY(1, 1) := p
 
  For NumTrials := 2 To MAX_TRIALS
    NumTrialsPlusOne := NumTrials + 1
 
    ; Handle limit cases first. Note that Probability array was zeroed above.
    If p = 0 Then
      PROBABILITY(NumTrials, 0) := 1
    ElseIf p = 1 Then
      PROBABILITY(NumTrials, NumTrials) := 1
    Else
      Mode := ROUND_DOWN((NumTrialsPlusOne) * p, 0)
 
      ; Start big.  It's all downhill from here.
      PROBABILITY(NumTrials, Mode) := 1.79e308 / NumTrialsPlusOne
 
      ; Calculate unscaled probabilities going up.
      NumSuccesses := Mode + 1
      Do While NumSuccesses < NumTrialsPlusOne
        PROBABILITY(NumTrials, NumSuccesses) := PROBABILITY(NumTrials, NumSuccesses - 1) * _
          (p * (NumTrials - NumSuccesses + 1) ) / (q * NumSuccesses)
        NumSuccesses := NumSuccesses + 1
      Loop

      ; Calculate unscaled probabilites going down.
      NumSuccesses := Mode - 1
      Do While NumSuccesses > -1
        PROBABILITY(NumTrials, NumSuccesses) := PROBABILITY(NumTrials, NumSuccesses + 1) * _
          (q * (NumSuccesses + 1) ) / (p * (NumTrials - NumSuccesses ))
        NumSuccesses := NumSuccesses - 1
      Loop
 
      ; Add them up, smallest to tallest.
      Total := 0
      LeftIdx := 0
      RightIdx := NumTrials
      Do While LeftIdx <= RightIdx
        If PROBABILITY(NumTrials, LeftIdx) < PROBABILITY(NumTrials, RightIdx) Then
          Total := Total + PROBABILITY(NumTrials, LeftIdx)
          LeftIdx := LeftIdx + 1
        Else
          Total := Total + PROBABILITY(NumTrials, RightIdx)
          RightIdx := RightIdx - 1
        EndIf
      Loop
 
      ; Normalize the probabilities.
      For NumSuccesses := 0 To NumTrials
        PROBABILITY(NumTrials, NumSuccesses) := PROBABILITY(NumTrials, NumSuccesses) / Total
      Next
 
  EndIf
 
  Next ; Num Trials
 
  ; That’s all folks!
  Initialized := 1
EndIf

